



FREE eBook

LEARNING

autoit

Free unaffiliated eBook created from
Stack Overflow contributors.

#autoit

Table of Contents

About	1
Chapter 1: Getting started with autoit	2
Remarks.....	2
Examples.....	2
Installation or Setup.....	2
Hello World.....	2
Chapter 2: Interaction with Excel	4
Examples.....	4
Simple writing to Excel.....	4
Credits	6

About

You can share this PDF with anyone you feel could benefit from it, downloaded the latest version from: [autoit](#)

It is an unofficial and free autoit ebook created for educational purposes. All the content is extracted from [Stack Overflow Documentation](#), which is written by many hardworking individuals at Stack Overflow. It is neither affiliated with Stack Overflow nor official autoit.

The content is released under Creative Commons BY-SA, and the list of contributors to each chapter are provided in the credits section at the end of this book. Images may be copyright of their respective owners unless otherwise specified. All trademarks and registered trademarks are the property of their respective company owners.

Use the content presented in this book at your own risk; it is not guaranteed to be correct nor accurate, please send your feedback and corrections to info@zzzprojects.com

Chapter 1: Getting started with autoit

Remarks

AutoIt is a lightweight freeware BASIC-like scripting language designed for automating the Windows GUI and general scripting.

It is especially useful for generating keystrokes and reading information from native Windows controls. Library functions are included for the automation of Microsoft Office applications and web browsers.

It can be downloaded from <https://www.autoitscript.com/site/autoit/>, a site which also contains a comprehensive documentation set.

Examples

Installation or Setup

AutoIt is intended for use on the Microsoft Windows operating system. It is compatible with versions from Windows XP onwards.

Download the installation utility from <https://www.autoitscript.com/site/autoit/downloads/>

If installing on a 64-bit version of Windows, the user is prompted to choose a 64- or 32-bit installation. The choice affects the compilation utilities that are installed alongside the script interpreter.

The installation utility will install into the given directory:

- The script interpreter
- A script compiler
- A script editor (SciTE)
- Help files
- Examples

Hello World

A simple AutoIt script showing some of the basic features of the language.

Note that semicolons introduce comments

Save this as "HelloWorld.au3"

```
; Include constants. In this case, for the MsgBox() function
#include <MsgBoxConstants.au3>

; Define and initialize the title of a dialogue box
; All AutoIt variables are loosely typed
; Local specifies the scope
```

```
Local $title
$title = "Example Window"


; Alternative method of implicit definition
; This time also indicating constant
Const $text = "Hello World"

; Third example definition
; Strings and integers defined in the same manner
$timeout = 30

; Create dialogue box using #included constants
MsgBox($MB_ICONINFORMATION, $title, $text, $timeout)

; Program end - Use of Exit is not mandatory
Exit
```

This will produce a simple output:-



Read Getting started with autoit online: <https://riptutorial.com/autoit/topic/6014/getting-started-with-autoit>

Chapter 2: Interaction with Excel

Examples

Simple writing to Excel

A basic script showing the fundamentals of how to create/open, write to, and save an Excel spreadsheet. Using the Autolt Excel library functions.

```
; Include definitions of the User Excel functions
#include <Excel.au3>

; Include Message Box codes for the error messages
#include <MsgBoxConstants.au3>

; Constants
$xlforce   = True           ; True = force new instance, False = use existing
$xlkeybrd  = True           ; True = allow keys,           False = block keys
$xlscreen  = True           ; True = live screen,         False = suppressed
$xlvisible = True           ; True = visible,           False = invisible

; Variables
$book      = 0              ; The workbook
$excel     = 0              ; The instance of Excel
$range     = 0              ; Range object within a sheet

; Create an instance of Excel and attach this script to it
; Use the constants defined above to control visibility of the Excel window
$excel = _Excel_Open($xlvisible, False, $xlvisible, $xlkeybrd, $xlforce)
If ($excel = 0) Then HandleError("Excel creation failed ")

; Create a new workbook with one worksheet within the opened instance
$book = _Excel_BookNew($excel, 1)
If ($book = 0) Then HandleError("Workbook creation failed ")

; Example write to a single cell in the current sheet - writes 42 to cell B2
$range = _Excel_RangeWrite($book, Default, 42, "B2")
If ($range = 0) Then HandleError("Write cell failed ")

; Example write to a block in the current sheet - writes to cells D4:E5
Local $xy[2][2] = [[1, 2], [3, 4]]
$range = _Excel_RangeWrite($book, Default, $xy, "D4:E5")
If ($range = 0) Then HandleError("Write cell failed ")

; Save the created workbook - without a given path this will go to the temporary
; directory. The argument 'TRUE' forces an overwrite of any existing file
$err = _Excel_BookSaveAs($book, "test.xlsx", $xlWorkbookDefault, True)
If ($err = 0) Then HandleError("Workbook save failed ")

; Close the workbook. The argument 'TRUE' will save any changes
$err = _Excel_BookClose($book, True)
If ($err = 0) Then HandleError("Workbook close failed ")

; Close the created instance of Excel
; First argument 'TRUE' saves changes to any open worksheets
; Second argument 'FALSE' prevents closure if the instance was not new at _Excel_Open()
```

```
$err = _Excel_Close($excel, True, False)
If ($err = 0) Then HandleError("Excel close failed ")

Exit                                     ; End of script

; Error handler function - very simple, just aborts
Func HandleError($message)
    MsgBox($MB_SYSTEMMODAL + $MB_ICONERROR, "Excel Test", $message & @error, 5)
    Exit
EndFunc
```

Read Interaction with Excel online: <https://riptutorial.com/autoit/topic/7872/interaction-with-excel>

Credits

S. No	Chapters	Contributors
1	Getting started with autoit	Bookeater , Chenmunka , Community
2	Interaction with Excel	Chenmunka